SIF5005 våren 2003: Maple-øving 2

Løsningsforslag.

1 Oppgave 1: Funksjoner, partiellderiverte, gradienter

Først tegner vi funksjonen fra oppgave 8 i hjemmeøving 2:

```
> f:=(x,y)->exp(-x^2)*(y^2+1);
```

$$f := (x,\, y) \to e^{(-x^2)}\,(y^2+1)$$

```
> plot3d(f(x,y),x=-2..2,y=-1..1);
```



Og så den andre funksjonen det ble spurt om:

```
> f:=(x,y)->x^3-x*y^2+y^3;
```

$$f := (x,\, y) \to x^3 - x\,y^2 + y^3$$

```
> D[1](f);
```

$$(x,\, y) \to 3\,x^2 - y^2$$

```
> D[2](f);
```

$$(x,\, y) \to -2\,x\,y + 3\,y^2$$

```
> region:=x=-1..1,y=-1..1:
> plot3d(f(x,y),region); fplot:=%:
```



Både funksjonen og de to partiellderiverte i ett plott:

```
> plots[display3d](
> fplot,
> plot3d(D[1](f)(x,y),region),
> plot3d(D[2](f)(x,y),region));
```



Kan du se hvilken som er hvilken av dem? (Nei, det er ikke lett!) Det blir litt lettere om vi fargelegger de deriverte (og skalerer dem ned litt):

```
> plots[display3d](
> fplot,
> plot3d(0.2*D[1](f)(x,y),region,color=red),
> plot3d(0.2*D[2](f)(x,y),region,color=blue));
```



Men mye lettere blir det å se på konturplott og gradienter:

```
> plots[contourplot](f(x,y),region,filled=true,levels=20);
> fcplot:=%:
```




```
> plots[gradplot](f(x,y),region); fgplot:=%:
```




```
> plots[display](fcplot,fgplot,scaling=constrained);
```



Vi kan til og med kombinere todimensjonale konturplott med tredimensjonale bilder:

```
>  xyplan:=plottools[transform]((x,y)->[x,y,0]):
>  plots[display3d](fplot,xyplan(fcplot),xyplan(fgplot),scaling=constrain
>  ed);
```



2 Oppgave 2: Fra hjemmeøving 3

Oppgave 6:

```
> f:=(x,y)->-x^3+x^2+3*x-x^2*y+y^2/2;
```

$$f := (x,\, y) \to -x^3 + x^2 + 3\,x - x^2\,y + \frac{1}{2}\,y^2$$

Spør Maple om største og minste verdi for f, uten bibetingelser {} og der variablene er {x,y} (se hjelpesiden for ***extrema***):

```
> extrema(f(x,y),{},{x,y},'p');
```

$$\{\frac{-3}{2},\, \frac{5}{2}\}$$

Variabelen p inneholder de kritiske punktene (tre i tallet):

```
> p;
```

$$\{\{x = \frac{-3}{2},\, y = \frac{9}{4}\},\, \{y = 1,\, x = 1\},\, \{y = 1,\, x = -1\}\}$$

Her er en tegning, hvor vi skulle ha alle de kritiske punktene innenfor:

```
> plots[contourplot](f(x,y),x=-2..2,y=-1..3,scaling=constrained,contour
> s=40,filled=true);
```



En kjapp liten liste over funksjonsverdiene i de kritiske punktene:

```
> for xy in p do
> print(subs(xy,'f(x,y)'=f(x,y)));
> end;
```

$$\mathrm{f}(\frac{-3}{2},\, \frac{9}{4}) = \frac{-45}{32}$$

$$\mathrm{f}(1,\, 1) = \frac{5}{2}$$

$$\mathrm{f}(-1,\, 1) = \frac{-3}{2}$$

Nytt konturplott, hvor jeg selv velger nivåene slik at jeg får med detaljer rund de kritiske punktene:

```
> ct:=[seq(0.1*i,i=-20..-10),seq(0.1*i,i=10..30)]:
> plots[contourplot](f(x,y),x=-2..2,y=-1..3,scaling=constrained,contours
> =ct,filled=true,grid=[50,50]);
```




```
> plot3d(f(x,y),x=-2..2,y=-1..3,axes=boxed,view=-3..3,orientation=[150,
> 70]);
```



3 Oppgave 3: Noen overraskelser med kritiske punkter

3.1 «One peak on a mountainside»

Kilde: Eksamen 75012 Matematikk 1B, 1991-05-25.

```
> f:=(x,y)->3*x*exp(y)-x^3-exp(3*y):  'f(x,y)'=f(x,y);
```

$$\mathrm{f}(x,\,y) = 3\,x\,e^{y} - x^{3} - e^{(3\,y)}$$

```
> extrema(f(x,y),{},{x,y},'p');
```

$$\{1\}$$

```
> p;
```

$$\{\{x = \mathrm{RootOf}(\_Z^{2} + \_Z + 1),\, y = \ln(-1 - \mathrm{RootOf}(\_Z^{2} + \_Z + 1))\},\, \{y = 0,\, x = 1\}\}$$

RootOf betyr en rot av polynomet gitt som argument. Vi vil gjerne ha det skrevet ut med rottegn:

```
> convert({%},radical);
```

$$\{\{\{y = 0,\, x = 1\},\, \{x = -\frac{1}{2} + \frac{1}{2}\,I\,\sqrt{3},\, y = \ln(-\frac{1}{2} - \frac{1}{2}\,I\,\sqrt{3})\}\}\}$$

Bare ett *reelt* kritisk punkt, ser det ut til. Det andre er komplekst.

```
> A=D[1,1](f)(1,0), B=D[1,2](f)(1,0), C=D[2,2](f)(1,0);
> subs(%,A*C-B^2);
```

$$A = -6,\, B = 3,\, C = -6$$

$$27$$

Det er et **lokalt maksimum**. Globalt? Nei! Med $y = 0$ kan vi la x gå mot $-\infty$ og få vilkårlig store verdier.

```
> region:=x=-2..2.5, y=-3..0.75:
> plots[contourplot](f(x,y),region,grid=[50,50],
> filled=true,contours=[seq(i,i=-8..-1),seq(0.1*i,i=-9..9),seq(i,i=1..5)
> ]);
```




```
> plot3d(f(x,y),region,grid=[50,50],
> color=exp(f(x,y)/3.5),style=patchcontour,axes=boxed,view=-10..5);
```



Det er en lokal fjelltopp på en fjellside. Når vi følger åsryggen ned fra fjelltoppen, kan vi gå uendelig langt uten å forlate denne åsryggen - og alltid nedoverbakke. Bakom er det en dal som alltid går oppover, men heller ikke den ender noe sted.

3.1.1 Hvordan fant de på dette?

Eksemplet er sikkert laget ved å starte med $3\,x\,y - x^3 - y^3$, som har et maksimum i [1, 1] og et sadelpunkt i origo. Deretter erstattes y med e^y, som flytter sadelpunktet ut til det uendelig fjerne.

```
> g:=(x,y)->3*x*y-x^3-y^3;
> plots[contourplot](g(x,y),x=-2..2,y=-1..2,filled=true,scaling=constra
> ined,contours=40,grid=[50,50]);
```

3.2 Twin peaks , no saddle

Jeg har hentet eksemplet fra en melding i Usenet:

From: cleary@zimmer.csufresno.edu (**Sean Cleary**) **Subject: Re: Crititical Points of Polynomials** Newsgroups: sci.math.research Date: 17 Mar 1998 21:35:35 GMT Organization: California State University, Fresno

Also see the problem "Two Mountains without a Valley", proposed and solved by Ira Rosenholz, *Mathematics Magazine*, Vol 60, No 1, Feb 1987 p.48, referenced in Anton's Calculus book, which gives an analytic solution.

```
> f:=(x,y)->-(x^2-1)^2-(x^2*y-x-1)^2:  'f(x,y)'=f(x,y);
```

$$f(x,\, y) = -(x^2 - 1)^2 - (x^2\,y - x - 1)^2$$

```
> extrema(f(x,y),{},{x,y},'p');
```

$$\{0\}$$

```
> p;
```

$$\{\{y = 0,\ x = -1\},\ \{x = 1,\ y = 2\}\}$$

(En nærmere undersøkelse viser at de begge er lokale maksimumspunkter. Vel, det er forresten opplagt, for funksjonsverdien i de kritiske punktene er 0, mens funksjonen ellers alltid er negativ! Burde det ikke da være et sadelpunkt et sted? Nei!

```
> region:=x=-2..2,y=-2..10:
> plot3d(f(x,y),region,
> grid=[50,50],orientation=[105,70],view=-5..0,
> color=exp(f(x,y)),style=PATCHCONTOUR,scaling=UNCONSTRAINED,axes=BOXED)
> ;
```



4 Oppgave 4: Flater i kule- og sylinderkoordinater

4.1 Kulekoordinater

Kulekoordinater er i Maple gitt som $(\rho,\ \theta,\ \phi)$. Man kan lage parametriske flater med disse koordinatene, eller plotte ρ som funksjon av θ og ϕ.

Her er figuren til oppgave 5 fra hjemmeøving 2:

```
> plot3d(1-cos(phi),theta=0..2*Pi,phi=0..Pi,coords=spherical,
> axes=normal,scaling=constrained);
```



Kanskje litt lettere å se om man skjærer bort halve flaten:

```
>  plot3d(1-cos(phi),theta=0..Pi,phi=0..Pi,coords=spherical,
>  axes=normal,scaling=constrained,orientation=[-30,55]);
```



Og kanskje sammenligner med kardioiden, som foreslått:

```
> plot([1-cos(theta),theta,theta=0..2*Pi],coords=polar,scaling=constrai
> ned);
```



4.2 Sylinderkoordinater

Sylinderkoordinater er i Maple gitt som (r, θ, z). Man kan lage parametriske flater med disse koordinatene, eller plotte r som funksjon av θ og z. Hvis vi vil tegne en flate der z er gitt som funksjon av r og θ, for eksempel $z = r^2$, må vi altså bruke parametrisk plott:

```
> plot3d([r,theta,r^2],r=0..1.5,theta=0..2*Pi,coords=cylindrical,axes=n
> ormal);
> pplot:=%:
```



Dette var et rotasjonslegeme om z-aksen. Vil vi lage det samme rotasjonslegemet om x-aksen eller y-aksen er det bare å ***transform***ere plottet vi nettop laget:

```
> xzbytt:=plottools[transform]((x,y,z)->[z,y,x]):
> yzbytt:=plottools[transform]((x,y,z)->[x,z,y]):
> plots[display3d](xzbytt(pplot),scaling=constrained,axes=normal);
```



Her kommer figuren til oppgave 4 fra hjemmeøving 2:

```
>  plots[display3d](xzbytt(pplot),yzbytt(pplot),axes=normal,scaling=cons
>  trained);
```